(2) 知識同化機構の一実現法

北上　始，麻生盛敏，国藤　進，宮地泰造，古川康一
（財団法人・新世代コンピュータ技術開発機構）

1. はじめに

人間の知的な精神活動を計算機システムに行なわせるためには、種々の計算機構を備えなければならないが、その一つとして、人間がもつ知識を組織的に蓄積管理する機構があげられる。これは、一般に、知識ベースの研究として知られている(1)。

我々は、この研究の一環として、知識獲得の分野に属する知識同化の実現法について検討している(2),(3)。知識同化とは、知識が蓄積されているデータベースに対して、ユーザの意図に合った知識を、入力(3),(4),(5)、または、作成する機能を意味する。

以下では、データベースに蓄積されている知識を、(i)ファクト(個々の事実)、(ii)ルール(一般的な規則)、(iii) IC(ファクトとルールについてのIntegrity Constraint)としてとらえ、データベースの形式を、関係データベースへの演繹的質問応答システムとしてとらえる。以後、このシステムを、論理データベースと呼ぶ。

演繹的機能として、SEQUELやQUELなどの言語を、サポートしている関係データベースは、知識として、主に、ファクトを対象にしていた。論理データベースは、本格的に、ルールを知識の対象にしている。

本稿では、知識ベースシステム向きの知識同化機構を、ルールの場合も含めて検討し、演繹的かつ帰納的推論機構を採用するという立場で、それを体系的に整理したので報告する。
さらに、そのいくつかを、Prologによりインプリメントしたので、その結果についても報告する。

2. 知識同化機構の構成

知識同化機構に重点をおいた論理データベースのアイデアは、Kowalskiらにより検討されている(6)。
本章では、そのアイデアに対する考察と、それに基づいて再構成された知識同化機構の概念構成について述べる。ただし、以下では、知識の形式としてホーン節を基本にする。

2.1 論理データベースの考察

Kowalskiらのアイデアでは、次のような手続きを必要としている。

A1) 論理データベースへの質問応答
A2) データベースから、入力知識の証明による知識同化の拒否
A3) データベースの冗長性除去
A4) 無矛盾なデータベースの管理
A5) データベースと独立な知識の同化
A6) 入力知識よりも有益な知識の生成と同化

A1)では、データベースに含まれているファクトとルールに対する演繹的質問応答を提供している。一般に、質問の形態は、存在限定(existential quantifier)及び全称限定(universal quantifier)が混在する命題$f(x_1,\cdots,x_n)$である。
$x_1,\cdots,x_n$は、それぞれ、どちらかの量限定を受ける変数である。本稿では、任意の質問は、どちらか1種類の量限定から成る命題とする。著者らは、既に、存在限定だけから成る命題(以下では、存在命題と呼ぶ)向きの証明を、demo述語と呼ばれるPrologインタプリタで実現している(3)。
全称限定だけから成る命題(以下では、全称命題と呼ぶ)の証明については、3章で、その実現方法を述べる。

以後、特に、存在命題を証明する述語を、existential-demo述語と呼び、全称命題を証明する述語を、universal-demo述語と呼ぶ。

A2)では、ある入力知識をデータベースに同化しようとするとき、データベースから、その入力知識を証明できるならば、その知識の同化を拒否している。その証明は、入力知識がファクトであれば、existential-demo述語により実現されるが、その知識がルールであれば、universal-demo述語により実現される。

A3)では、データベースに入力知識を挿入した際に、冗長性が生ずることがあるので、それを除去している。入力知識がデータベースに同化されていることを前提として、データベースから任意の知識を取り出し、その知識が残りのデータベースから証明されるとき、取り出された知識は、冗長な知識であるという。このときの証明過程においても、上記、2種のdemo述語が利用される。

ところで、冗長性除去の処理は、ユーザによって、その処理を希望する場合と、そうでない場合がある。例えば、良く利用される知識が、非常に時間のかかる演繹によって得られるのであれば、ユーザは、直接その知識を、データベース内に蓄積するかも知れない。

A4)では、データベースに入力知識を挿入した際に、意味的な矛盾が生ずると、その入力知識の同化を拒否している。データベースの意味的な矛盾性は、ICにより発見される。一般に、ICは、ファクトやルールを制限するメタな知識である。

A5)では、A2)からA4)までの判定により、入力知識が同化できるとわかったとき、入力知識は、データベースと独立であるとしている。明らかに、独立な入力知識は、データベースから証明されない。

A6)では、データベースに仮説を付加することによって、独立な入力知識を証明できれば、その仮説を、入力知識の代わりに同化している。

本稿では、この問題を、帰納的なモデル推論(7)の問題としてとらえる。

2.2 知識同化機構の概念構成

図1に、知識ベースシステム向きの知識同化機構を組織的に再構成した概念図である。



図1. 知識同化機構の概念図

前節の考察から、2種のdemo述語が、知識同化の基本機能として、重要であることがわかる。これは、演繹推論の機構である。これらのdemo述語は、Prologのプログラミング言語をオブジェクトとするメタ言語である。オブジェクト言語とメタ言語を融合して、利用する。両言語の融合については、(8)の発表でその詳細が述べられるが、データベースのホスト言語に対応する概念である。

図1のassimilationは、知識の同化

に関する核部分の機能であるが、前節のA6)の機能を含めていない。ICの判定及び冗長性除去の機能が含まれている。

evolutionは、入力知識をもっと高度な知識としてデータベースに同化したいときに利用される。これを、帰納的なモデル推論により実現する。

dissimilationは、同化された知識が、その後の業務で不要になったとき、利用される。詳細は、文献(9)を、参照されたい。

複数のassimilationが、一つの単位として、実行される場合、途中段階で、ICを満さないことがある。transaction controlは、その便宜をはかるために必要である。

deductive question_answeringは、メタ述語によるデータベースへの質問応答機能を提供する。

その他に、データベースのルールに関する包含関係及び同等性を証明する機能があげられるが、これらは、他の分野で作成されたデータベースを有効利用するときに重要な機能である。

3. 演繹的推論機構

本章では、知識同化で利用される演繹推論機構のexistential_demo述語と、universal_demo述語の実現法について述べる。

3.1 存在命題の証明

図2に、この命題を証明するためのexistential_demo述語の実現例を示す。命題は、以下の形を仮定する。

1) Head :- Goals.
2) Head.

ただし、"Head"は、1つの述語であり、"Goals"は、述語の論理和及び論理積から成る。また、"Goals"に、カットオペレータ、否定述語"not(·)"、システムの組み込み述語"system(·)"を許す。

図2のexistential_demo述語は、2引数をもつ。第1引数は、使用するデータベースの名前DBを与え、第2引数は、証明すべき存在命題を指定する。図中のdemo述語は、Prologインタプリタとして知られている(3)。また、clause_DB述語は、述語Pを"Head"とする"Goals" QをDBから探索する述語である。

```
existential_demo(DB,(Head:-Goals)):-!,
     (demo(DB,Head,_);not(demo(DB,Goals,_))).
existential_demo(DB,Head):-demo(DB,Head,_).


demo(DB,true,Cut):-!,true.
demo(DB,!,Cut).
demo(DB,!,cut).
demo(DB,(P;Q),Cut):-!,
     (demo(DB,P,Cut);demo(DB,Q,Cut)).
demo(DB,(P,Q),Cut):-!,
     demo(DB,P,Value),
     (Value==cut,Cut=cut;demo(DB,Q,Cut)).
demo(DB,P,Cut):-
     clause_DB(DB,P,Q),demo(DB,Q,Cut),
     (Cut==cut,!,fail;true).
demo(DB,not(P),Cut):-!,not(demo(DB,P,Cut)).
demo(DB,P,Cut):-system(P).


clause_DB(DB,P,Q):-
     X=..[DB,(P:-Q)],X.
clause_DB(DB,P,Q):-
     X=..[DB,P],Q=true,X.
```

図2. 存在命題の証明用述語

3.2 全称命題の証明

ここでは、以下の前提をおく。

1) 命題は、3.1節の形式にしたがう。したがって、命題の中に現われる変数は、全称限定付きである点だけが違う。
2) 命題中の変数は、すべて、有限領域を対象とした変数とする。すなわち、命題は、有限のファクトの上に定義されているとする。

以上の前提から、全称命題"Head:-Goals"を、次のアルゴリズムにより証明できる。全称命題が"Head"のときは、"Head:-true"と考えて、このアルゴリズムを適用すれば良い。

U1) Prologインタプリタのdemo述語を利用して、"Goals"を満足するすべてのインスタンスIを、"Goals"から取り出す。
U2) Iが空のとき、"Head"の述語名がデータベース中に存在すれば、証明結果を真とする。
U3) Iが空でないとき、Iのすべてのインスタンスに対し、"Head"が真であれば、証明結果を真とする。
U4) U2)及びU3)のどちらの条件も満さないとき、証明結果を、偽とする。

図3に、この全称命題を証明するためのuniversal_demo述語の実現例を示す。この述語も、existential_demo述語と同様に、2引数をもつ。

```
universal_demo(DB,(Head:-Goals)):-
     select_variable_list(Goals,X),
     setof(X,demo(DB,Goals,_),Set),
     length(Set,N),N=\=0,!,
     prove_all_instance(DB,X,Set,Head).
universal_demo(DB,not(Head)):-
     not(demo(DB,Head,_)).
universal_demo(DB,Clause):-
     (Clause=(Head:-Goals);Clause=Head),
     \+(\+(clause_DB(DB,Head,_))).


prove_all_instance(DB,X,[Elem|Set],Head):-
     ground(Elem,X,Head,Ground_Head),!,
     demo(DB,Ground_Head,_),!,
     prove_all_instance(DB,X,Set,Head).
prove_all_instance(DB,X,[],Head).
```

図3. 全称命題の証明用述語

"select_variable_list"は、"Goals"のすべての変数をXに渡す述語である。"setof"では、"Goals"がもつすべてのインスタンスをさがし、変数Setに出力している。"prove_all_instance"は、U3)に相当する処理を行う述語である。

4. 帰納推論機構

図1のevolutionの機構を、帰納的なモデル推論の問題としてとらえる。このモデル推論の考え方は、Shapiroの研究(7)を参考にしている。本章では、このモデル推論の研究を考察し、知識同化への適用方法について述べる。

4.1 モデル推論

図4に、Shapiroが紹介した増殖的なモデル推論アルゴリズムを示す。このアルゴリズムは、ユーザにより、真または偽の観測文αが与えられている。観測構文αは、ファクトであり、Vは、"true"または"false"の真偽値である。ここでは、αとVの組を、観測事実F_nと呼ぶ。推測される仮説の集合は、Prologでプログラムされた手続き(ルールとファクトの集合)である。

L_kから、ファクトαの証明を、3.1節で述べたdemo述語により達成できる。矛盾探索アルゴリズムの実現は、Prologのユーザにとって、非常に簡単である。すなわち、図2に示されている6番目のdemo述語で、ゴール文の最後に、PのインスタンスとS_{false}のファクトαを照合する機構を導入すれば良い。もし、その照合に失敗したら、"P:-Q"は、反ばく仮説になる。

洗練演算子については、帰納関数の理論により構成される。

図の2つのwhile文が満足しないとき、次の観測構文の読み込み待ちになるが、このとき、T_kは、$T_k \nvdash \alpha_{false}$かつ$T_k \vdash \alpha_{true}$を満足している。ただし、$\alpha_{false} \in S_{false}$, $\alpha_{true} \in S_{true}$である。

4.2 知識同化への適用方法

入力知識をデータベースに同化しようとしたとき、入力知識よりも有用な仮説を推論するために、モデル推論アルゴリズムが適用される。これは、ユーザの意志にしたがう。

入力知識がファクトのとき、その知識を、本アルゴリズムで作成された仮説

$S_{false} = \{\square\}$, $S_{true} = \{\}$とする。
$L_0 = \{\square\}$とし、この□に、'false'をマークする。

repeat.
　次の観測事実 $F_n = \langle \alpha, V \rangle$ を読み込み、α を S_V に付け加える。
　repeat.
　　while〈推測された仮説の集合 L_k から、観測文 $\alpha \in S_{false}$ を証明できる〉do.
　　矛盾探索アルゴリズムを適用し、反ばく仮説に'false'をマークし、その仮説を L_k から除去する。
　　while〈推測された仮説の集合 L_k から、観測文 $\alpha_i \in S_{true}$ を証明できない〉do.
　　$k = k+1$ とする。すなわち、α_i を証明できる仮説を、洗練演算子で作成し、それを L_k に付け加え L_{k+1} とする。
　　until〈上記、whileループの条件のどちらも成立しない〉
　　output L_k を出力する。
forever.

図4. 増殖的なモデル推論アルゴリズム

におきかえ同化することができる。
もし、この入力知識に関し、さらに多くの情報を、ユーザが持っている場合は、さらに、正確な仮説を、生成することができる。

入力知識がルールのとき、データベースに、それを同化後、そのルールを拡張する観測事実を与えることによって、さらに有用な仮説を生成することができる。

さて、モデル推論の探索空間を狭ばめるために、次の情報を与えなければならない。

1) 述語の名前と、引数の数
2) 引数の構造（例えば、ストリング、リスト）
3) 引数の入/出力仕様
4) 組み込み述語の名前
5) その他

これらは、データベースの辞書情報としてとらえることができる。したがって、これらを管理する辞書機能が必要である。

5. 知識同化アルゴリズムの改善

本章では、2章で考察したKowalskiのアイデアに基づき、図1で位置づけたassimilationのアルゴリズムについて述べる。本アルゴリズムでは、独立性の概念の定義として、Nicolasが採用した定義を使う(11)。また、本アルゴリズムに関し、次の主な前提をおく。

1) 閉世界仮説を仮定する。
2) 現在、対象としているデータベースは、無矛盾である（ICも無矛盾であるとする）。
3) データベースのデータ（ルール、ファクト、IC）は、Prologで記述される。したがって、述語変数は、すべて、全称限定付きである。これにより、否定述語 not(P) の中に記述される述語Pは、存在限定付きの変数である。

以上から、assimilationのアルゴリズムは、次のように整理される。

A1) データベースから入力知識を証明できれば、知識の同化を実施しない。
A2) データベースから入力知識の否定を証明できるならば、知識の同化を実施しない。
A3) データベースに入力知識が同化されたことを前提に、データベースの冗長性を除去する。ただし、冗長性除去の処理は、ユーザの意志にまかせる。
A4) データベースに入力知識が同化されたことを前提に、データベー

スからICの否定を証明できれば、そのデータベースは、矛盾するので、データベースを最初の状態にもどす（A3)で冗長性が除去されていることがある)。

A5) A1)からA4)までのいずれの手続きに対しても、入力知識が知識同化の対象になっているならば、入力知識を、データベースに挿入する。

本アルゴリズムでは、Kowalskiが述べている独立性の概念を、Nicolasの概念としてとらえ、その手続きを採用している。それによると、A1)及びA2)が共に満足しないとき、入力知識は、データベースに独立であると見做すことができる。

6. 知識同化プログラムの実行例

付録Iに、実現された知識同化プログラムの中で、assimilateの部分を示す。ただし、5章のA4)でのデータベース回復処理は、今後の問題としている。

本章では、よく知られている「積み木」の問題を例に、その実行例を示す。

6.1 問題の設定

図5に「積み木」の配置図を示す。床aの上に、b, c, ･･･, h, iの積み木を使って、タワーが作られているとする。積み木fは、長方形の形をしており、それ以外は、正方形をしている。その他に、jとkの積み木が手中にあるとする。jは、長方形で、kは、正方形である。



図5. 積み木の配置図

図6に、その配置関係を、Prologで表現した例を示す。積み木は、"block"という述語で表現され、積み木の積み重ね関係は、"on"という述語で表現されている。

本データベース中のファクトは、述語"rectangular_block", "square_block", "floor", "on"であり、ルールとしては、述語"block", "tower"がある。データベース名は、"build"と名づけられている。

```
/* blocks in the rectangular shape */

floor(a).
rectangular_block(f).
rectangular_block(j).

/* blocks in the square shape */

square_block(b).
square_block(c).
square_block(d).
square_block(e).
square_block(g).
square_block(h).
square_block(i).
square_block(k).


/* block */

block(X):-
      square_block(X);
      rectangular_block(X);
      floor(X).


/* on(X,Y) : X is on Y */

on(b,a).
on(d,a).
on(c,b).
on(f,c).
on(e,d).
on(f,e).

on(g,a).
on(h,g).
on(i,h).


/* tower(X,Y) : The tower consists of block X */
/*                  on top of tower Y.        */

tower(X,Y):-towerl(X,Y),Y\==[].

towerl(X,[]):-floor(X).
towerl(X,[Y|Z]):-block(X),on(X,Y),towerl(Y,Z).
```

図6. 積み木の配置関係のデータベース

ICとして、次の制限をおく。ただし、ICの名前は、"ic"である。

1) "floor(a)"は、決して削除されない。
2) "rectangular_block"は、2本以上の"tower"で支えられる。
3) 2)の"tower"には、"square_block"だけから成る"tower"が1本以上存在する。

図7に、この関係を示す。

```
floor(a).

ge(N,2):-
      rectangular_block(X),
      setof(build,(tower(X,Y),ne(Y,[])),[Y],S),
      ne(S,[]),length(S,N).

ge(N2,1):-
      rectangular_block(X),
      setof(build,(tower(X,Y),ne(Y,[])),[Y],S),
      ne(S,[]),length(S,N1),ge(N1,2),
      setof(build,(tower(X,V),on(X,W),
            square_block(W),ne(V,[])),[V],S2),
      length(S2,N2).

le(N,4):-tower(X,Y),length(Y,N).
```

図7. 積み木データベースのIC

以上から、次の問題を解いてみる。
P1) 長方形と正方形の積み木から作られるコーナーを認識する述語"corner"の同化。
P2) 積み木の組み立てと分解に関するICの問題。
P3) 述語"above"の同化。

付録IIに、その実行結果を示す。本システムでは、ICのチェックを実施しない知識同化の処理が入っている。すなわち、ICとして[]を与えることにより疑似的なtransaction controlを実現している。また、evolutionの処理では、ICのチェックを簡単のため省略している。

7. おわりに

本稿では、知識ベースシステム向きの知識同化機構について、入力知識がルールである場合を含めて検討し、それを、Prologでインプリメントした結果を示した。

知識同化の基本機構として、存在命題と全称命題を証明するための演繹的推論機構について述べた。また、知識同化の一環として、さらに有効な知識を同化するために、帰納的推論機構について述べた。

今後の課題として、以下の項目について検討する予定である。
1) 本知識同化処理では、1回の処理で入力できる知識を、1つの節に限定したが、これを複数の節としたときの実現方法。
2) 冗長性除去の一般化と高速化。
3) universal_demo述語の拡張*。これは、全称命題を、skolem化し、"Goals"の部分を、データベースに付加して"Head"の証明を(demo述語で)実施すれば良い。
4) 帰納的なモデル推論の高速化。
5) 事実によるIC(メタ知識)の帰納的なモデル推論の実現。
6) データベースのTrigger及びActionの検討。
7) データベース間の同等性を証明する問題。
8) 並列演算機構の導入による高速化。

[謝辞]
本研究の機会を与えて頂いた当財団法人・新世代コンピュータ技術開発機構の淵一博研究所所長と、有益な討論をして頂いた第2研究室の自然言語グループの皆様に深く感謝致します。

[参考文献]
(1) 淵一博：問題解決と推論機構，情報処理学会誌 Vol.19 No.10 (1978).
(2) 志村正道：知識の獲得と学習，信学技報 AL80-46 (1980).

* これは、グラウンドインスタンスをもたない知識を、データベースで蓄積管理するときに必要である。

(3) 宮地，国藤，北上，古川，竹内，横田：論理データベース向きの知識同化方式の一提案，京大数解研講究録「モデル表現とその構築に関する理論と実際の研究」(1983)．
(4) Bowen, K.A., Kowalski, R.A. : Amalgamating Language and Meta-Language in Logic Programming, School of Computer and Information Sciences, University of Syracuse (1981).
(5) Kowalski, R.A. : Logic as a Database Language, Department of Computing, Imperial College (1981).
(6) Kowalski, R.A.: Logic for Problem Solving, Elsevier North Holland, Inc., pp. 239－246 (1979).
(7) Shapiro, E.Y. : Inductive Inference of Theories From Facts, Technical Report 192, Department of Computer Science, Yale University (1981).
(8) 国藤，麻生，竹内，宮地，北上，横田，安川，古川：Prologによる対象知識とメタ知識の融合とその応用，情報処理学会研究会資料「知識工学と人工知能」(1983)．
(9) 北上，宮地，国藤，古川：知識ベースシステムKAISERの構想，情報処理学会第26回全国大会(1983)．
(10) Eswaran, K.P. and Chamberlin, D.D.: Functional Specification of a Subsystem for Database Integrity, Proc. 1st VLDB Conf. (1975).
(11) Nicolas, J.M. : Logic for Improving Integrity Checking in Relational Data Bases, Acta Informatica (1982).
(12) Chamberlin, D.D. etal. : A Unified Approach to Data Definition, Manipulation, and Control, IBM J RES. DEVELOP. (1976).

付録 I. assimilation のプログラム例

```
/* (A1) : Is "Input" derivable from "Current_kb" ?  */
assimilate(Current_kb,IC_name,Input):-
      demonstrate(Current_kb,Input),
      nl,write('* Reject. ---> The Input_Knowledge "'),
      write(Input),write('" '),nl,
      write(' is derivable from the Current_Knowledge_Base.'),nl,
      nl,write('* Reject the Input_Knowledge ? (y/n) '),ttyflush,
      ttyget0(X),ttyskip(31),X=:=121.

/* (A2) : Is "Nega(not(P))" derivable from "Current_kb" ? */
assimilate(Current_kb,IC_name,not(P)):-
      demonstrate(Current_kb,P),
      nl,write('* Reject the Input_Knowledge ? (y/n) '),ttyflush,
      ttyget0(X),ttyskip(31),X=:=121.

/* (A3) : Does "Input" imply information already implicitly
                                        contained in "Current_kb" ? */
assimilate(Current_kb,IC_name,Input):-
      remove_redundancy,
      (open_generator(Current_kb);close_generator,fail),
      generate_temp_kb(Current_kb,Input,[P1,P2,P3,Information]),
      demonstrate(temp_kb,Information),
      close_generator,
      nl,write('* The Knowledge "'),write(Information),write('" '),
      nl,write(' was removed from Current_Knowledge_Base.'),nl,
      delete_knowledge(Current_kb,[P1,P2,P3]).

/* (A4) : Is ["Current_kb" + "Input"] inconsistent with "IC" ? */
assimilate(Current_kb,IC_name,Input):-
      IC_name\==[],
      dict(IC_name,[Q1,Q2,Maxptr]),
      get_integrity_constraint(IC_name,IC,Q1),
      negate(IC,Not_ic),
      insert_knowledge(Current_kb,Input,[P1,P2,P3]),
      (set_backtrace(IC_name);free_backtrace,fail),
      (demonstrate(Current_kb,Not_ic);
      delete_knowledge(Current_kb,[P1,P2,P3]),fail),
      free_backtrace,
      delete_knowledge(Current_kb,[P1,P2,P3]),
      nl,write('* The Input_Knowledge "'),write(Input),write('" '),
      nl,write(' is inconsist with the Integrity_Constraint.'),nl,
      nl,write('* Reject the Input_Knowledge ? (y/n) '),ttyflush,
      ttyget0(X),ttyskip(31),(X=:=110,
      nl,write('* Refine the Current_Knowledge_Base ? (y/n) '),
      ttyflush,ttyget0(Y),ttyskip(31),(Y=:=121,
      refinement(Current_kb,IC_name,Input);true);true).

/* (A5) : "Input" is logically independent from "Current_kb" -->
                                                    Insert "Input". */
assimilate(Current_kb,IC_name,Input):-
      insert_knowledge(Current_kb,Input,Ptr),
      nl,write('* The Input_Knowledge "'),write(Input),write('" '),
      write(' was assimilated.'),nl.
```

付録 II. 積み木データベースの実行例

```
% ?- assim(build,ic,corner(f,[c,b,a])).

* Eliminate Redundancy ? (y/n) n

* The Input_Knowledge "corner(f,[c,b,a])"  was assimilated.

 4638 msec.

yes
% ?- assim(build,ic,(corner(X,Y):-tower(X,Y))).

* Eliminate Redundancy ? (y/n) y.

* The Knowledge "corner(f,[c,b,a])"
 was eliminated from Current_Knowledge_Base.

* The Input_Knowledge "(corner(_0,_1):-tower(_0,_1))"  was assimilated.

 13836 msec.

yes
```

```
% ?- existential_demo(build,corner(X,Y)).

corner(b,[a])  ;

corner(c,[b,a])  ;

corner(d,[a])  ;

corner(e,[d,a])  ;

corner(g,[a])  ;

corner(h,[g,a])  ;

corner(i,[h,g,a])  ;

corner(f,[c,b,a])  ;

corner(f,[e,d,a])  ;

no
% ?- evolution(build,corner(X,Y)).

* Initialize parameter(y/n)? n.

Next fact(sentence,true/false) or end ? corner(b,[a]),false.
Checking fact(s)...
Error: wrong solution corner(b,[a]). diagnosing...

Query: block(b)? y.

Query: towerl(a,[])? y.

Query: towerl(b,[a])? y.

Query: tower(b,[a])? y.
Error diagnosed: (corner(b,[a]):-tower(b,[a])) is false.
Listing of corner(X,Y):

Checking fact(s)...no error found.

Next fact(sentence,true/false) or end ? corner(f,[e,d,a]),true.
Checking fact(s)...
Error: missing solution corner(f,[e,d,a]). diagnosing...
Error diagnosed: corner(f,[e,d,a]) is uncovered.

Searching for a cover to corner(f,[e,d,a])...
Refining: (corner(X,Y):-true)
Refining: (corner(X,[Y|Z]):-true)
Refining: (corner(X,Y):-tower(X,U))
Checking: (corner(X,Y):-tower(X,Y))
Refuted: (corner(b,[a]):-tower(b,[a]))
Refining: (corner(X,[Y,Z|U]):-true)
Refining: (corner(X,[Y|Z]):-tower(X,V))
Refining: (corner(X,Y):-tower(X,Y))
Checking: (corner(X,Y):-tower(X,Y),rectangular_block(X))
Found clause: (corner(X,Y):-tower(X,Y),rectangular_block(X))
    after searching 9 clauses.
Listing of corner(X,Y):
    (corner(X,Y):-tower(X,Y),rectangular_block(X)).

Checking fact(s)...no error found.

Next fact(sentence,true/false) or end ? corner(i,[h,g,a]),false.
Checking fact(s)...no error found.

Next fact(sentence,true/false) or end ? end.
```

```
* Purge Facts(y/n) ? y.

* Eliminate Redundancy ? (y/n) n.

* Eliminate rejected theory? (y/n) y.

* Construction of the Knowledge "corner"
 has been finished.  6232 msec.

yes
% ?- existential_demo(build,corner(X,Y)).

corner(f,[c,b,a])  ;

corner(f,[e,d,a])  ;

no
% ?- assim(build,[],on(j,f)).

* Eliminate Redundancy ? (y/n) n.

* The Input_Knowledge "on(j,f)"  was assimilated.

 96 msec.

yes
% ?- assim(build,ic,on(j,i)).

* Eliminate Redundancy ? (y/n) n.

* The Input_Knowledge "on(j,i)"  was assimilated.

 9196 msec.

yes
% ?- existential_demo(build,corner(X,Y)).

corner(f,[c,b,a])  ;

corner(f,[e,d,a])  ;

corner(j,[f,c,b,a])  ;

corner(j,[f,e,d,a])  ;

corner(j,[i,h,g,a])  ;

no
% ?- assim(build,ic,on(k,j)).

* Eliminate Redundancy ? (y/n) n.

* By the contradiction backtracing,detected the following knowledge.
 ---> "(le(_0,4):-tower(_1,_2),length(_2,_0))"

* The Input_Knowledge "on(k,j)"
 is inconsist with the Integrity_Constraint.

 10486 msec.

yes
% ?- dissim(build,[],on(j,i)).

* The Knowledge "on(j,i)"
 was deleted from Current_Knowledge_Base.

yes
% ?- dissim(build,ic,on(j,f)).

* The Knowledge "on(j,f)"
 was deleted from Current_Knowledge_Base.

yes
```

```
% ?- dissim(build,[],on(f,c)).

* The Knowledge "on(f,c)"
 was deleted from Current_Knowledge_Base.

yes
% ?- dissim(build,ic,on(f,e)).

* The Knowledge "on(f,e)"
 was deleted from Current_Knowledge_Base.

yes
% ?- existential_demo(build,corner(X,Y)).

no
```

```
% ?- assimilate(build,ic,(above(X,Y):-on(X,Z),on(Z,Y))).

* The Input_Knowledge "(above(_0,_1):-on(_0,_2),on(_2,_1))"  was assimilated.

yes
% ?- existential_demo(build,above(X,Y)).

above(c,a)  ;

above(f,b)  ;

above(e,a)  ;

above(f,d)  ;

above(h,a)  ;

above(i,g)  ;

no
% ?- evolution(build,above(X,Y)).

* Initialize parameter(y/n)? n.

* The Knowledge_base "solutions" was created on Core_Space.

Next fact(sentence,true/false) or end ? above(b,a),true.
Checking fact(s)...
Error: missing solution above(b,a). diagnosing...

Query: on(a,a)? n.
Error diagnosed: above(b,a) is uncovered.

Searching for a cover to above(b,a)...
Refining: (above(X,Y):-true)
Refining: (above(X,Y):-on(X,U))
Checking: (above(X,Y):-on(X,Y))
Found clause: (above(X,Y):-on(X,Y))
    after searching 3 clauses.
Listing of above(X,Y):
    (above(X,Y):-on(X,U),on(U,Y)).
    (above(X,Y):-on(X,Y)).

Checking fact(s)...no error found.

Next fact(sentence,true/false) or end ? above(i,a),true.
Checking fact(s)...
Error: missing solution above(i,a). diagnosing...

Query: on(h,a)? n.

Query: on(i,a)? n.
Error diagnosed: above(i,a) is uncovered.
```

```
Searching for a cover to above(i,a)...
Refining: (above(X,Y):-true)
Refining: (above(X,Y):-on(X,U))
Refining: (above(X,Y):-on(X,U),on(U,W))
Refining: (above(X,Y):-on(X,U),above(U,W))
Checking: (above(X,Y):-on(X,U),above(U,Y))
Found clause: (above(X,Y):-on(X,U),above(U,Y))
    after searching 11 clauses.
Listing of above(X,Y):
    (above(X,Y):-on(X,U),on(U,Y)).
    (above(X,Y):-on(X,Y)).
    (above(X,Y):-on(X,U),above(U,Y)).

Checking fact(s)...no error found.

Next fact(sentence,true/false) or end ? end.

* Purge Facts(y/n) ? y.

* The Knowledge_base "solutions" was droped on Core_Space.

* Eliminate Redundancy ? (y/n) y.

* The Knowledge "(above(_0,_1):-on(_0,_2),on(_2,_1))"
 was eliminated from Knowledge_Base "build" .

* Eliminate rejected theory? (y/n) y.

* Construction of the Knowledge "above"
 has been finished.  12162 msec.

yes
% ?- existential_demo(build,above(X,Y)).

above(c,a)  ;

above(f,a)  ;

above(f,b)  ;

above(e,a)  ;

above(f,a)  ;

above(f,d)  ;

above(h,a)  ;

above(i,a)  ;

above(i,g)  ;

above(b,a)  ;

above(d,a)  ;

above(c,b)  ;

above(f,c)  ;

above(e,d)  ;

above(f,e)  ;

above(g,a)  ;

above(h,g)  ;

above(i,h)  ;

no
```